



POLITECNICO DI MILANO

3° Laboratorio

Fondamenti di Automatica

SCILAB e SCICOS

Analisi in Frequenza

Stabilità dei Sistemi Lineari

Analisi di Sensitività

Giovanni Vannozzi --- anno accademico 2010-2011

Analisi in Frequenza Sistemi Lineari

- L'analisi in frequenza dei Sistemi Lineari Tempo Invarianti è svolta mediante la nozione fondamentale di **RISPOSTA IN FREQUENZA**;
- Un segnale sinusoidale di frequenza assegnata può essere univocamente rappresentato da un numero complesso, detto **FASORE**, il cui **modulo coincide con l'ampiezza** del **segnale** e l'**argomento** con la **fase del segnale stesso**;
- La risposta di un sistema Lineare Tempo Invariante ad un Ingresso Sinusoidale, *se tale ingresso non coincide con un modo naturale del sistema*, è costituita dalla **somma** di una **componente transitoria**, combinazione lineare dei **modi** del sistema, e di **una componente a regime** caratterizzata da una **sinusoide della stessa pulsazione ω dell'ingresso**, con fasore uguale al prodotto del fasore dell'ingresso per il **valore della Funzione di Trasferimento $G(s)$** valutata in corrispondenza di **$s=j\omega$** . \Rightarrow **Teorema fondamentale dell'analisi armonica**


Analisi in Frequenza Sistemi Lineari

- la **Risposta in Frequenza** $G(j\omega)$ è una **funzione complessa** nella **variabile reale** ω (**pulsazione**) che **identifica** in modo completo le *proprietà filtranti* del sistema avente *Funzione di Trasferimento* $G(s)$;
- il **modulo** $|G(j\omega)|$ definisce il *guadagno* del *sistema* alla *pulsazione* ω , cioè l'**amplificazione** se $|G(j\omega)| > 1$, oppure l'**attenuazione** se $|G(j\omega)| < 1$, subita da una **sinusoide** di **pulsazione** ω nel **passaggio** dall'ingresso all'uscita attraverso il sistema;
- l'**argomento** $\angle G(j\omega)$ rappresenta lo **sfasamento** del **sistema** alla **pulsazione** ω , cioè il **ritardo di fase** se $\angle G(j\omega) < 0$, o l'**anticipo di fase** se $\angle G(j\omega) > 0$, subito da una **sinusoide** alla **pulsazione** ω nel suo **trasferimento** dall'ingresso all'uscita del sistema;
- di solito la **Risposta in Frequenza** è rappresentata **graficamente** da **due diagrammi cartesiani** per il **modulo** e per la **fase**, noti come **diagrammi di Bode**, oppure con **un solo diagramma** detto **diagramma polare**, il cui opportuno ampliamento definisce il **diagramma di Nyquist**.

Analisi in Frequenza Sistemi Lineari

Assegnato un Sistema Lineare si vuole determinare la **risposta** in **frequenza** mediante la definizione, ovvero come **immagine** del **semiasse immaginario positivo $j\omega$** tramite la funzione di rete $G(j\omega)$, al fine di tracciare i grafici relativi :

- ❖ del modulo $|G(j\omega)|$ o $|G(j\omega)|_{dB}$ in “**scala lineare**” avendo come ascissa, in “**scala logaritmica**” la pulsazione ω ;
- ❖ della fase $\angle G(j\omega)$, in “**scala lineare**” in gradi, avendo come ascissa, in “**scala logaritmica**”, la pulsazione ω .



$$G(s) \Big|_{s=j\omega} = G(j\omega)$$

Analisi in Frequenza Sistemi Lineari

La funzione **horner** serve per **valutare un polinomio $P(x)$** nel **punto x_0** ovvero **$P(x_0)$** ; se x_0 è un **vettore**, **horner** restituisce un **vettore** le cui **componenti** sono i **valori** di **$P(x)$** **calcolato** in corrispondenza delle **componenti** di x_0 .

--> **pol=poly([1 2 3],'x','c');** $\Rightarrow p(x) = 1 + 2x + 3x^2$

--> **xo=3;**

--> **polxo=horner(pol,xo)** $\Rightarrow p(xo) = 1 + 2 \cdot 3 + 3 \cdot 3^2$
polxo = 34.

--> **pol=poly([1 2 3],'x','c');** $\Rightarrow p(x) = 1 + 2x + 3x^2$

--> **xv=[-1 0 1 2 3];**

--> **polxv=horner(pol,xv)**
polxv = 2. 1. 6. 17. 34.

Analisi in Frequenza Sistemi Lineari

La funzione **horner** permette, poi, di calcolare il valore assunto da una *frazione* algebrica, data come **rapporto** di due **polinomi** $N(s)/D(s)$, nel **punto** s_0 ovvero $N(s_0)/D(s_0)$; quando s_0 è un **vettore**, **horner** fornisce un **vettore** le cui **componenti** sono i **valori** della frazione $N(s)/D(s)$, **calcolata** in corrispondenza delle **componenti** di s_0 .

```
--> num=poly([1 2],'s','c');
```

$$\Rightarrow \text{num}(s) = 1 + 2s$$

```
--> den=poly([1 1 1],'s','c');
```

$$\Rightarrow \text{den}(s) = 1 + s + s^2$$


```
--> fra=num/den;
```

```
--> sv=[-2 -1 -0.5 0 1];
```

```
--> frasv=horner(pol,sv)
```

```
frasv =
```

```
-1. -1. 0. 1. 1.
```


$$\text{fra} = \frac{1 + 2s}{1 + s + s^2}$$

Analisi in Frequenza Sistemi Lineari

La funzione **plot2d** , con l'opzione **logflag** consente il tracciamento dei grafici nelle **scale semilogaritmiche**, nelle **scale logaritmiche** oppure nelle **scale lineari**.

```
--> t=-1:0.01:5;
```

```
--> y=exp(-t);
```

```
--> plot2d(t,y,logflag='nl'),xgrid
```

```
--> xtitle('grafico della funzione exp(-t)')
```

logflag='nn' ascisse e ordinate in scala lineare


logflag='nl' ascisse scala lineare, ordinate logaritmica

logflag='ln' ascisse scala logaritmica, ordinate lineare

logflag='ll' ascisse e ordinate in scala logaritmica

Analisi in Frequenza Sistemi Lineari


Si ottiene la seguente rappresentazione grafica di: $y = e^{-t}$



Analisi in Frequenza Sistemi Lineari

Si consideri il **sistema** del **secondo ordine**, con uno **zero aggiuntivo**, definito dalla “**Funzione di Trasferimento**” $G(s)$ di seguito riportata:

$$G(s) = \frac{(1 + 10s)}{(1 + s) \cdot (1 + 100s)}$$



Si desidera rappresentare **graficamente** l'andamento del **modulo** e della **fase** della funzione di rete $G(j\omega)$ relativi alla **risposta in frequenza** del sistema nell'**intervallo** di **pulsazione** $10^{-4} \leq \omega \leq 10^2$.

Analisi in Frequenza Sistemi Lineari

Il listato nel codice dell'ambiente Scilab è il seguente:

```
--> num=poly([1 10],'s','c');           ⇒ num=(1+10s)
--> den1=poly([1 1],'s','c');           ⇒ den1=(1+s)
--> den2=poly([1 100],'s','c');         ⇒ den2=(1+100s)
--> den=den1*den2;
--> w=logspace(-4,2,500);
--> RF=horner(num,%i*w)./horner(den,%i*w);
--> mRF=abs(RF);
--> fRF=phasemag(RF);
--> subplot(211),plot2d(w,mRF,logflag='ln'),xgrid
--> subplot(212),plot2d(w,fRF,logflag='ln'),xgrid
```

Il denominatore poteva determinarsi subito con l'istruzione:

```
--> den=poly([1 1],'s','c')*poly([1 100],'s','c');
```


Analisi in Frequenza Sistemi Lineari

Un listato, in ambiente Scilab, alternativo ma sempre equivalente al precedente, viene mostrato dal codice seguente:

```
--> s=poly(0,'s');  
--> num=1+10*s;           ⇒ num=(1+10s)  
--> den=(1+s)*(1+100*s);  ⇒ den=(1+s)·(1+100s)  
--> giesse=num/den;  
--> w=logspace(-4,2,500);  
--> RF=horner(giesse,%i*w);  
--> mRF=abs(RF);  
--> fRF=phasemag(RF);  
--> subplot(211),plot2d(w,mRF,logflag='ln'),xgrid  
--> subplot(212),plot2d(w,fRF,logflag='ln'),xgrid
```

Analisi in Frequenza Sistemi Lineari

Si ottengono i grafici dei moduli e delle fasi relativi alla **risposta in frequenza** nell'intervallo di pulsazione considerato



Modulo
espresso
in scala
lineare

Fase
espressa
in gradi

Analisi in Frequenza Sistemi Lineari

La grandezza fisica A viene espressa in **decibel**, A_{dB} , allorché viene calcolato il suo **valore** con la seguente relazione:

$$A_{dB} = 20 \cdot \log_{10} |A|$$

Pertanto, nel caso della **funzione di rete** $G(j\omega)$, si deve intendere:

$$|G(j\omega)|_{dB} = 20 \cdot \log_{10} |G(j\omega)|$$


Il **diagramma** di **Bode** è, quindi, determinato dal seguente codice:

```
-->mRFdb=20*log10(mRF);  
-->fRF=phasemag(RF);  
-->subplot(211),plot2d(w,mRFdb,logflag='ln'),xgrid  
-->subplot(212),plot2d(w,fRF,logflag='ln'),xgrid
```

Analisi in Frequenza Sistemi Lineari

I diagrammi di Bode rappresentano il **modulo** di $G(j\omega)$ in **dB**

Modulo
riportato
in decibel
su scala
lineare



Fase
espressa
in gradi

Analisi in Frequenza Sistemi Lineari

Siano assegnati i tre **sistemi** del **secondo ordine**, con uno **zero aggiuntivo**, definiti dalle **Funzione di Trasferimento** $G(s)$ di seguito riportate:

$$G_1(s) = \frac{(1+10s)}{(1+s) \cdot (1+0,1s)}$$

$$G_2(s) = \frac{(1+1s)}{(1+10s) \cdot (1+0,1s)}$$

$$G_3(s) = \frac{(1+0,1s)}{(1+1s) \cdot (1+10s)}$$


Si vuole rappresentare **graficamente** l'andamento dei **moduli** e delle **fasi** delle tre **funzione di rete** $G_1(j\omega)$, $G_2(j\omega)$ e $G_3(j\omega)$, afferenti alla **risposta in frequenza** dei relativi sistemi, nell'intervallo di **pulsazione** $10^{-2} \text{ r/s} \leq \omega \leq 10^2 \text{ r/s}$.

Lo scopo è quello di valutare il ruolo espletato dallo zero.

Analisi in Frequenza Sistemi Lineari

```
-->num1=poly([1 10],'s','c');
-->num2=poly([1 1],'s','c');
-->num3=poly([1 0.1],'s','c');
-->den1=poly([1 1],'s','c')*poly([1 0.1],'s','c');
-->den2=poly([1 10],'s','c')*poly([1 0.1],'s','c');
-->den3=poly([1 10],'s','c')*poly([1 1],'s','c');
-->w=logspace(-2,2,2000);
-->RF1=horner(num1,%i*w)./horner(den1,%i*w);
-->RF2=horner(num2,%i*w)./horner(den2,%i*w);
-->RF3=horner(num3,%i*w)./horner(den3,%i*w);
-->mRF1=20*log10(abs(RF1));fRF1=phasemag(RF1);
-->mRF2=20*log10(abs(RF2));fRF2=phasemag(RF2);
-->mRF3=20*log10(abs(RF3));fRF3=phasemag(RF3);
-->subplot(211),plot2d(w,[mRF1',mRF2',mRF3'],logflag='ln'),xgrid
-->subplot(212),plot2d(w,[fRF1',fRF2',fRF3'],logflag='ln'),xgrid
```


Analisi in Frequenza Sistemi Lineari



$$|G_1(j\omega)|_{dB}$$

$$|G_2(j\omega)|_{dB}$$

$$|G_3(j\omega)|_{dB}$$


$$\angle G_1(j\omega)$$

$$\angle G_2(j\omega)$$

$$\angle G_3(j\omega)$$

Analisi in Frequenza Sistemi Lineari

Sempre ricorrendo all'utilizzo della **funzione horner**, si desidera calcolare e plottare il **Modulo** e la **Fase** della **Funzione di Rete** $G(j\omega)$ afferente la **Risposta in Frequenza** del Sistema Lineare caratterizzato dal **ritardo** $T_0 = 0,2$ e dalla seguente **Funzione di Trasferimento** $G(s)$:

$$G(s) = \frac{(1-5s)}{(1+s) \cdot (1+2s)^2} \cdot e^{-0,2s}$$

$$G(j\omega) = \frac{(1-j5\omega)}{(1+j\omega) \cdot (1+j2\omega)^2} \cdot e^{-j0,2\omega}$$

Analisi in Frequenza Sistemi Lineari


Il listato del codice in ambiente SCILAB è il seguente:

```
--> num=poly([1 -5], 's', 'c');  
--> den=poly([1 1], 's', 'c')*poly([1 2], 's', 'c')^2;  
--> d=-0.2; j=sqrt(-1); (ritardo: d = T0 = -2)  
--> w=logspace(-4, 2, 1000);  
--> RF=(horner(num, j*w)./horner(den, j*w)).*exp(d*j*w);  
--> mRF=abs(RF);  
--> fRF=phasemag(RF);  
--> subplot(211), plot2d(w, mRF, logflag='ll'), xgrid  
--> subplot(212), plot2d(w, fRF, logflag='ln'), xgrid
```

$$G(j\omega) = \frac{(1 - j5\omega)}{(1 + j\omega) \cdot (1 + j2\omega)^2} \cdot e^{-j0,2\omega}$$

Analisi in Frequenza Sistemi Lineari


Grafici relativi alla **Risposta in Frequenza** di $G(j\omega) \cdot e^{j\omega T_0}$



Ritardo
 $T_0 = -2$

Analisi in Frequenza Sistemi Lineari

Grafici relativi alla **Risposta in Frequenza** di $G(j\omega) \cdot e^{j\omega T_0}$



Ritardo
 $T_0 = -2$

Analisi in Frequenza Sistemi Lineari

Listato relativo al confronto delle due Funzioni di Rete


```
-->j=sqrt(-1); d= -0.2; (ritardo: d = T0 = -2)
-->s=poly(0,'s'); w=logspace(-3,2,1000);
-->GS=(1-5*s)/((1+s)*(1+2*s)^2);
-->RF=horner(GS,j*w);
-->RFR=horner(GS,j*w).*exp(j*d*w);
-->mRF=abs(RF);mRFR=abs(RFR);
-->fRF=phasemag(RF); fRFR=phasemag(RFR);
-->subplot(211),plot2d(w,[mRF',mRFR'],logflag='ln'),xgrid
-->subplot(212),plot2d(w,[fRF',fRFR'],logflag='ln'),xgrid
```

$$G(j\omega) = \frac{(1 - j5\omega)}{(1 + j\omega) \cdot (1 + j2\omega)^2}$$

$$G_R(j\omega) = G(j\omega) \cdot e^{-j0,2\omega}$$

Analisi in Frequenza Sistemi Lineari

Il confronto fra le Risposte in Frequenza è dato dal grafico:



Analisi in Frequenza Sistemi Lineari

Si considerino le due seguenti **Funzioni** di **Trasferimento**:

$$G_1(s) = \frac{1}{1+s} \quad G_2(s) = \frac{1}{1+s} \cdot e^{-0,1s}$$


Si desiderano determinare le relative **Risposte in Frequenza** al fine di rappresentarle nel **piano complesso** della **variabile “s”** come **curve punteggiate** dalla **pulsazione “ω”**.

Listato per il tracciamento del “Diagramma Polare”:

```
--> num=1; den=poly([1 1], 's', 'c'); d=-0,1;  
--> w=logspace(-3,6,1000);  
--> RF1=horner(num, %i*w)./horner(den, %i*w);  
--> RF2=RF1.*exp(d*%i*w);  
--> plot(real(RF1),imag(RF1),real(RF2),imag(RF2)),xgrid
```



Analisi in Frequenza Sistemi Lineari

Si ottiene il relativo Diagramma polare mostrato in figura



Analisi in Frequenza Sistemi Lineari

Zoom del diagramma polare ($10 < \omega < 1000$) mostra l'effetto del ritardo



Analisi in Frequenza Sistemi Lineari

La risposta in frequenza si ottiene anche, nota la $G(s)$ del sistema lineare, direttamente con l'istruzione **repfreq**, i cui parametri sono la $G(s)$, la **frequenza minima** f_{\min} , la **frequenza massima** f_{\max} dell'intervallo di frequenze volute e l'incremento o **step** da usare per “**spazzolare**” l'intervallo delle frequenze assegnato $[f_{\min}, f_{\max}]$.

L'istruzione **repfreq** restituisce due parametri che sono:

frq = il vettore le cui componenti definiscono i valori della **frequenza** utilizzati per determinare la **risposta in frequenza**;

repf = il vettore le cui componenti rappresentano i numeri complessi $G(j\omega)$ calcolati per i valori di $j\omega$ ottenuti dalla relazione: $j\omega = j \cdot 2\pi \cdot \mathbf{frq}$.

Analisi in Frequenza Sistemi Lineari

Utilizziamo l'istruzione **repfreq** per ottenere la **risposta** in **frequenza** del sistema lineare definito dalla **funzione di trasferimento**:


$$G(s) = \frac{(1 + 10s)}{(1 + s) \cdot (1 + 100s)}$$

```
--> num=poly([1 10], 's', 'c');           ⇒ num=(1+10s)
--> den=poly([1 101 100], 's', 'c');       ⇒ den=(1+101s +100s²)
--> giesse=syslin('c', num, den);
--> [frq, repf]=repfreq(giesse, 0.00001595, 15, 0.001);
--> w=2*%pi*frq;
--> subplot(211), plot2d(w, abs(repf), logflag='ln'), xgrid
--> subplot(212), plot2d(w, phasemag(repf), logflag='ln'), xgrid
```

Analisi in Frequenza Sistemi Lineari

Si ottiene, in **SCILAB**, lo stesso insieme di grafici della **risposta in frequenza** già ottenuto facendo ricorso alla istruzione **horner**.

Modulo
espresso
in scala
lineare



Analisi in Frequenza Sistemi Lineari

Con l'istruzione: `[db,phi]=dbphi(repf)`

si ottiene il **modulo** $|G(j\omega)|$, espresso in **decibel**, e la **fase** $\angle G(j\omega)$, espressa in **gradi**, della **Risposta in Frequenza** del **sistema lineare** con **funzione di trasferimento** $G(s)$.


Esempio: il seguente listato, in ambiente SCILAB,

```
--> num=poly([1 10],'s','c');  
--> den=poly([1 101 100],'s','c');  
--> giesse=syslin('c', num, den);  
--> [frq, repf]=repfreq(giesse, 0.00001595, 15, 0.001);  
--> w=2*%pi*frq;  
--> [db, phi]=dbphi(repf);  
--> subplot(211),plot2d(w, db, logflag='ln'),xgrid  
--> subplot(212),plot2d(w, phi, logflag='ln'),xgrid
```

Analisi in Frequenza Sistemi Lineari

restituisce i grafici del modulo, espresso in dB, e della fase, espressa in gradi, della risposta in frequenza.

Modulo
riportato
in decibel
su scala
lineare



Fase
espressa
in gradi
su scala
lineare

Relazione fra Risposta nel Tempo e Risposta in Frequenza

È interessante ricercare la **relazione** fra la **Banda Passante** e la **Velocità** di **Risposta** per un **Sistema Lineare**.

Per semplicità e per una maggiore evidenza si considerano **tre Sistemi Lineari** del **Primo ordine** e privi di **zeri**, aventi **guadagno unitario** e definiti mediante le relative **Funzioni di Trasferimento**, come di seguito riportato:

$$G_1(s) = \frac{1}{(1+s)} \quad G_2(s) = \frac{1}{(1+5s)} \quad G_3(s) = \frac{1}{(1+10s)}$$


Relazione fra Risposta nel Tempo e Risposta in Frequenza

Le seguenti righe di codice, in ambiente SCILAB:

```
--> w=logspace(-2,2,500);  
--> RF1=horner(1./poly([1 1],'s','c'),%i*w);  
--> RF2=horner(1./poly([1 5],'s','c'),%i*w);  
--> RF3=horner(1./poly([1 10],'s','c'),%i*w);  
--> m1=abs(RF1); m2=abs(RF2); m3=abs(RF3);  
--> t=0:0.05:60;  
--> giesse1=syslin('c', 1, poly([1 1],'s','c'))  
--> giesse2=syslin('c', 1, poly([1 5],'s','c'));  
--> giesse3=syslin('c', 1, poly([1 10],'s','c'));  
--> y1=csim('step', t, giesse1);  
--> y2=csim('step', t, giesse2);  
--> y3=csim('step', t, giesse3);  
--> subplot(211),plot2d(w,[m1' m2' m3'],logflag='ll'),xgrid  
--> subplot(212),plot2d(t,[y1' y2' y3']),xgrid
```

Relazione fra Risposta nel Tempo e Risposta in Frequenza

Consentono di ottenere i seguenti grafici:




Relazione fra Risposta nel Tempo e Risposta in Frequenza

Misuriamo il **modulo** e lo **sfasamento** della **risposta** in **frequenza**, ad una data **pulsazione** ω , mediante la conoscenza della **risposta** nel **tempo**.

```
--> num=poly([1 10], 's', 'c');  
--> den=poly([1 101 100], 's', 'c');  
--> gies=syslin('c', num, den);  
--> w1=0.03;w2=0.4;w3=2;  
--> t1=0:0.05:800; t2=0:0.05:100; t3=0:0.01:20;  
--> u1=sin(w1*t1); u2=sin(w2*t2); u3=sin(w3*t3);  
--> y1=csim(u1,t1,gies);y2=csim(u2,t2,gies);y3=csim(u3,t3,gies);  
--> subplot(311),plot(t1,u1,t1,y1),xgrid  
--> subplot(312),plot(t2,u2,t2,5*y2),xgrid  
--> subplot(313),plot(t3,u3,t3,y3),xgrid  
--> subplot(313),plot(t3,u3,t3,10*y3),xgrid
```

Relazione fra la Risposta nel Tempo e la Risposta in Frequenza



$$y_1$$

$$5 \cdot y_2$$

$$10 \cdot y_3$$


Relazione fra la Risposta nel Tempo e la Risposta in Frequenza

Le **uscite** y_2 ed y_3 sono rappresentate amplificate, rispettivamente, dei valori **5** e **10** al fine di meglio procedere alla determinazione dei seguenti **parametri**:

L'**ampiezza** Y della **sinusoide** d'**uscita** a **regime**, tenendo in conto che le **uscite** y_2 ed y_3 vengono visualizzate con **zoom** con **fattore**, rispettivamente, di valore **5** e **10**.

La **distanza temporale** D tra i due **massimi** corrispondenti dell'**ingresso** e dell'**uscita**, ovvero fra i due **punti** relativi allo **attraversamento** dell'**asse** dei **tempi** corrispondenti e successivi ai due massimi; tale **distanza** D è da prendersi col **segno +** se la **uscita** y_i **anticipa** l'**ingresso** u_i , col **segno -** se l'**uscita** y_i **ritarda** sull'**ingresso** u_i .

Relazione fra la Risposta nel Tempo e la Risposta in Frequenza



Relazione fra la Risposta nel Tempo e la Risposta in Frequenza

I risultati delle misurazioni effettuate sono di seguito riportati

Pulsazione $\omega_1 = 0,03$ r/s (periodo $T_1 = 209$ s):

$$|Y_1| = 0,35 \Rightarrow |G(j0,03)| = 0,35/1 = 0,35$$

$$D = -32 \Rightarrow \angle G(j0,03) = (-32/209) \cdot 360^\circ = -56^\circ$$

Pulsazione $\omega_2 = 0,4$ r/s (periodo $T_2 = 15,7$ s):

$$|Y_2| = 0,5/5 = 0,1 \Rightarrow |G(j0,4)| = 0,1/1 = 0,1$$

$$D = -1,6 \Rightarrow \angle G(j0,4) = (-1,6/15,7) \cdot 360^\circ = -36^\circ$$

Pulsazione $\omega_3 = 2$ r/s (periodo $T_3 = 3,14$ s):

$$|Y_3| = 0,5/10 = 0,05 \Rightarrow |G(j2)| = 0,05/1 = 0,05$$

$$D = -0,6 \Rightarrow \angle G(j2) = (-0,6/3,14) \cdot 360^\circ = -68^\circ$$

I Diagrammi di BODE

L'istruzione idonea al tracciamento dei **diagrammi** di **BODE** si presenta in variegate strutture adeguate ai diversi utilizzi:

bode(sys, fmin, fmax, step, 'comments')

bode([sys1;sys2], fmin, fmax, step, 'comments')

bode(sys, frq, 'comments')

bode(frq, db, phi, 'comments')

bode(frq, repf, 'comments')

sys, frq, db, phi, repf, fmin, fmax, step

conservano il significato già esplicitato in precedenza

comments è una stringa con cui si definiscono i commenti

fmin, fmax, step e **comments** sono opzionali

I Diagrammi di BODE

Si devono tracciare i **diagrammi di BODE** delle **Funzioni** di **Trasferimento** per ognuno dei valori dei **parametri** dati:

$$G_1(s) = \frac{1}{1 + sT} \quad T = 1 \quad G_2(s) = \frac{1}{1 + sT} \quad T = -1$$

$$G_3(s) = \frac{1 + s\tau}{1 + sT} \quad \tau = 0,5 \quad T = 1 \quad G_4(s) = \frac{1 + s\tau}{1 + sT} \quad \tau = -0,5 \quad T = 1$$

$$G_5(s) = \frac{1}{\omega_n^2 + 2\xi\omega_n s + s^2} \quad \omega_n = 1 \quad \xi = 0,1$$
$$G_6(s) = \frac{1}{\omega_n^2 + 2\xi\omega_n s + s^2} \quad \omega_n = 1 \quad \xi = 1,1$$


I Diagrammi di BODE

```
--> giesse1=syslin('c',1,poly([1 1],'s','c'));  
--> giesse2=syslin('c',1,poly([1 -1],'s','c'));  
--> scf(1),bode([giesse1;giesse2],0.001,100)
```


```
--> giesse3=syslin('c',poly([1 0.5],'s','c'),poly([1 1],'s','c'));  
--> giesse4=syslin('c',poly([1 -0.5],'s','c'),poly([1 1],'s','c'));  
--> scf(2),bode([giesse3;giesse4],0.001,100)
```

```
--> giesse5=syslin('c',1,poly([1 0.2 1],'s','c'));  
--> giesse6=syslin('c',1,poly([1 2.2 1],'s','c'));  
--> giesse7=syslin('c',1,poly([1 0.02 1],'s','c'));  
--> scf(3),bode([giesse5;giesse6],0.001,100)
```

I Diagrammi di BODE



I Diagrammi di BODE



$$\tau = 0,5$$

$$T = 1$$

$$\tau = -0,5$$

$$T = 1$$


$$\tau = 0,5$$

$$T = 1$$

$$\tau = -0,5$$


$$T = 1$$

I Diagrammi di BODE




I Diagrammi di BODE

--> `bode([giesse5;giesse7],0.01,1)`




Verifica STABILITÀ con SCICOS

Schema per la **simulazione**, in **ambiente SCICOS**, della **risposta** al **gradino** relativa alla **verifica della stabilità** dei due sistemi lineari del **secondo ordine** ($\omega_n = 1; \xi = 0,1$ --- $\omega_n = 1; \xi = 1,1$)




Diagrammi di Bode – verifica stabilità

Le risposte al gradino unitario, corrispondenti ai diagrammi di BODE dei due sistemi del 2° ordine, sono di seguito riportate:



Risposta in Frequenza – Diagramma Polare

Si considerino le seguenti tre **Funzioni di Trasferimento**:




Si desiderano determinare le relative **Risposte in Frequenza** al fine di rappresentarle nel **piano complesso** della **variabile** “s” come **curve punteggiate** nella **pulsazione** “ ω ”.

Risposta in Frequenza – Diagramma Polare

Il relativo listato in ambiente **SCILAB** è il seguente:

```
-->num1=poly([1 2],'s','c');  
-->num2=poly([1 5],'s','c');  
-->num3=poly([1 10],'s','c');  
-->den=poly([1 0.1],'s','c')*poly([1 1],'s','c');  
-->w=logspace(-3,6,5000);  
-->RF1=horner(num1,%i*w)./horner(den,%i*w);  
-->RF2=horner(num2,%i*w)./horner(den,%i*w);  
-->RF3=horner(num3,%i*w)./horner(den,%i*w);  
-->plot(real(RF1),imag(RF1),real(RF2),imag(RF2),...),xgrid
```

Risposta in Frequenza – Diagramma Polare



$$\tau_1 = 2$$


$$\tau_2 = 5$$

$$\tau_3 = 10$$


Risposta in Frequenza – Diagramma Polare

Si considerino le seguenti tre **Funzioni di Trasferimento**:


$$G_4(s) = \frac{(1 + 2s)}{(1 + 0,1s) \cdot (1 + s)}$$



$$G_5(s) = \frac{(1 + 0,2s)}{(1 + 0,1s) \cdot (1 + s)}$$



$$G_6(s) = \frac{(1 + 0,05s)}{(1 + 0,1s) \cdot (1 + s)}$$




Si desiderano determinare le relative **Risposte in Frequenza** al fine di rappresentarle nel **piano complesso** della **variabile** “**s**” come **curve punteggiate** nella **pulsazione** “ **ω** ”.

Risposta in Frequenza – Diagramma Polare



Il relativo listato in ambiente **SCILAB** è il seguente:

```
-->num4=poly([1 2],'s','c');  
-->num5=poly([1 0.2],'s','c');  
-->num6=poly([1 0.05],'s','c');  
-->den=poly([1 0.1],'s','c')*poly([1 1],'s','c');  
-->w=logspace(-3,6,5000);  
-->RF4=horner(num4,%i*w)./horner(den,%i*w);  
-->RF5=horner(num5,%i*w)./horner(den,%i*w);  
-->RF6=horner(num6,%i*w)./horner(den,%i*w);  
-->plot(real(RF4),imag(RF4),real(RF5),imag(RF5),...),xgrid
```

Risposta in Frequenza – Diagramma Polare



I Diagrammi di NYQUIST



Facendo tendere R all'infinito e ε a zero, si ottiene una curva Γ che circonda tutto il semipiano destro

Il criterio di stabilità di Nyquist utilizza i diagrammi di Nyquist, cioè diagrammi polari tracciati per $-\infty < \omega < +\infty$, essendo: $G(-j\omega) = G^*(j\omega)$, in cui: $G^*(j\omega)$ è il complesso coniugato di $G(-j\omega)$.

Criterio di stabilità di Nyquist: $N = P_d \rightarrow$ Poli di $G(s)$ ubicati nel semipiano destro

I Diagrammi di NYQUIST

Si determini il diagramma di Nyquist come immagine $G(j\omega)$ della funzione di trasferimento $G(s)$ allorché $s=j\omega$ descrive l'intero asse immaginario ed essendo $G(s) \rightarrow 0$ quando $s \rightarrow \infty$


Si noti l'effetto introdotto dallo zero positivo, sito quindi nel semipiano destro della variabile complessa s .

Il listato in ambiente SCILAB è il seguente:

```
-->num=poly([2 -2],'s','c');  
-->den=poly([1 1],'s','c')*poly([1 0.5],'s','c');  
-->giesse=num/den;  
-->w=-250:0.05:250;
```

oppure: $w=\text{linspace}(-250,250,15000)$

```
-->RF=horner(giesse,%i*w);  
-->plot(real(RF),imag(RF)),xgrid
```




$$G_1(s) = \frac{2 \cdot (1-s)}{(1+s) \cdot (1+0.5s)}$$

Si ottiene, in tal modo, il diagramma di Nyquist della $G(s)$.

I Diagrammi di NYQUIST

Immagine di $G_1(j\omega)$ per $-\infty < \omega < +\infty$



I Diagrammi di NYQUIST

L'istruzione atta al tracciamento dei **diagrammi** di **NYQUIST** si presenta in forma simile a quella afferente i diagrammi di Bode:

nyquist(sys, fmin, fmax, step, 'comments')

nyquist([sys1;sys2], fmin, fmax, step, 'comments')

nyquist(sys, frq, 'comments')

nyquist(frq, db, phi, 'comments')

nyquist(frq, repf, 'comments')

sys, frq, db, phi, repf, fmin, fmax, step

conservano il significato già esplicitato in precedenza

comments è una stringa con cui si definiscono i commenti

fmin, fmax, step e **comments** sono opzionali

fmin, **assume valori negativi** onde tracciare $G(-j\omega) = G^*(j\omega)$

I Diagrammi di NYQUIST

Si ricavi di nuovo il diagramma di Nyquist della funzione di trasferimento $G_1(s)$, già in precedenza analizzata, applicando direttamente una delle istruzioni specifiche appena esaminate.

Il sistema reazionato negativamente è definito dalla seguente funzione di trasferimento ad anello aperto:


$$G_1(s) = \frac{2 \cdot (1-s)}{(1+s) \cdot (1+0,5s)}$$

Il listato in ambiente SCILAB è il seguente:

```
-->num=poly([2 -2],'s','c');  
-->den=poly([1 1],'s','c')*poly([1 0.5],'s','c');  
-->giesse=syslin('c',num,den);  
-->nyquist(giesse,-300,300,0.01,'zero positivo')
```

Si ottiene, in tal modo, il diagramma di Nyquist della $G_1(s)$.

I Diagrammi di NYQUIST



$$\begin{aligned} z_1 &= 1 \\ p_1 &= -1 \\ p_2 &= -2 \end{aligned}$$

I Diagrammi di NYQUIST

Si traccino i **diagrammi di NYQUIST** delle **Funzioni** di **Trasferimento** per ognuno dei valori dei **parametri** dati:

$$G_1(s) = \frac{1}{1 + sT} \quad T = 1 \quad G_2(s) = \frac{1}{1 + sT} \quad T = -1$$

$$G_3(s) = \frac{1 + s\tau}{1 + sT} \quad \tau = 0,5 \quad T = 1 \quad G_4(s) = \frac{1 + s\tau}{1 + sT} \quad \tau = -0,5 \quad T = 1$$

$$G_5(s) = \frac{1}{\omega_n^2 + 2\xi\omega_n s + s^2} \quad \omega_n = 1 \quad \xi = 0,1$$
$$G_6(s) = \frac{1}{\omega_n^2 + 2\xi\omega_n s + s^2} \quad \omega_n = 1 \quad \xi = 1,1$$


I Diagrammi di NYQUIST

```
--> giesse1=syslin('c',1,poly([1 1],'s','c'));  
--> giesse2=syslin('c',1,poly([1 -1],'s','c'));  
--> subplot(211),nyquist(giesse1,-1000,1000)  
--> subplot(212),nyquist(giesse2,-100,1000)
```

```
--> giesse3=syslin('c',poly([1 0.5],'s','c'),poly([1 1],'s','c'));  
--> giesse4=syslin('c',poly([1 -0.5],'s','c'),poly([1 1],'s','c'));  
--> scf(1),subplot(211)nyquist(giesse3,-1000,1000)  
--> subplot(212)nyquist(giesse4,-1000,1000)
```

```
--> giesse5=syslin('c',1,poly([1 0.2 1],'s','c'));  
--> giesse6=syslin('c',1,poly([1 2.2 1],'s','c'));  
--> scf(2),subplot(211),nyquist(giesse5,-1000,1000)  
--> subplot(212),nyquist(giesse6,-1000,1000)
```


I Diagrammi di NYQUIST



$T = 1$

$T = -1$

I Diagrammi di NYQUIST




$$T = 1$$

$$\tau = 0,5$$

$$T = 1$$

$$\tau = -0,5$$

I Diagrammi di NYQUIST



$$\omega_n = 1$$

$$\xi = 0,1$$


$$\omega_n = 1$$

$$\xi = 1,1$$

I Diagrammi di NYQUIST

Si determini il diagramma di Nyquist della seguente funzione di trasferimento ad anello aperto:

$$G(s) = \frac{(1+s^2)}{(2-s) \cdot (2+s) \cdot (4+s)}$$




In ambiente SCILAB, il listato di seguito riportato:

```
-->s=poly(0,'s');  
-->num=(1+s^2);  
-->den=(2-s)*(2+s)*(4+s);  
-->giesse=syslin('c',num,den);  
    oppure: -->giesse=syslin('c',num/den)  
-->nyquist(giesse)
```

Consente di determinare il diagramma di Nyquist della $G(s)$.

I Diagrammi di NYQUIST



$$\mu = 1/16$$

$$z_1 = +j$$

$$z_2 = -j$$

$$p_1 = +2$$


$$p_2 = -2$$

$$p_3 = +4$$

I Diagrammi di NYQUIST

Si determini il diagramma di Nyquist della seguente funzione di trasferimento ad anello aperto:

$$H(s) = \frac{(1 + s^2)}{(s - 2) \cdot (s + 2) \cdot (s + 4)}$$




In ambiente SCILAB, il listato di seguito riportato:

```
-->s=poly(0,'s');
-->num=(1+s^2);
-->den=(s-2)*(s+2)*(s+4);
-->accaesse=syslin('c',num,den);
    oppure: -->accaesse=syslin('c',num/den)
-->nyquist(accaesse)
```

Consente di determinare il diagramma di Nyquist della $H(s)$.

I Diagrammi di NYQUIST



$$\mu = -1/16$$

$$z_1 = +j$$

$$z_2 = -j$$

$$p_1 = +2$$


$$p_2 = -2$$

$$p_3 = +4$$

Stabilità Sistemi Lineari Reazionati

Si consideri il sistema lineare descritto dalla Funzione di Trasferimento:

$$G(s) = \frac{1}{s \cdot (1 + s)}$$



Si devono tracciare i diagrammi di BODE e di NYQUIST della Risposta in Frequenza relativa alla Funzione di Trasferimento $G(s)$.

Si consideri lo schema a blocchi del sistema avente reazione unitaria, di cui $G(s)$ rappresenta la Funzione di Trasferimento ad ANELLO APERTO.

Si discuta l'applicabilità dei Criteri di NYQUIST e di BODE per l'ANALISI di STABILITÀ del Sistema RETROAZIONATO.

Stabilità Sistemi Lineari Reazionati

Listato A

```
-->den=poly([0 1],'s','c')*poly([1 1],'s','c');  
-->giesse4=syslin('c',1,den);  
-->bode(giesse4)
```

Listato B


```
-->den=poly([0 1],'s','c')*poly([1 1],'s','c');  
-->giesse4=syslin('c',1,den);  
-->[frq,repf]=repfreq(giesse4);  
-->bode(giesse4,frq)
```

Listato C

```
-->den=poly([0 1],'s','c')*poly([1 1],'s','c');  
-->giesse4=syslin('c',1,den);  
-->[frq,repf]=repfreq(giesse4);  
-->[db,phi]=dbphi(rep);  
-->bode(frq,db,phi)
```

Stabilità Sistemi Lineari Reazionati

Diagrammi di BODE relativi ai listati **A**, **B** e **C**



A

B

C

Stabilità Sistemi Lineari Reazionati

Verifichiamo che il diagramma polare associato alla risposta in frequenza $G(j\omega)$ presenta, relativamente alla parte reale, un asintoto verticale per $\omega \rightarrow 0$.

$$G(s) = \frac{1}{s \cdot (1 + s)} \quad \longrightarrow \quad G(j\omega) = \frac{1}{j\omega \cdot (1 + j\omega)}$$

$$G(j\omega) = \frac{1}{j\omega - \omega^2} = \frac{-\omega^2 - j\omega}{\omega^2 + \omega^4} = -\frac{\omega - j}{\omega + \omega^3}$$

$$\lim_{\omega \rightarrow 0} \operatorname{Re}[G(j\omega)] = -\lim_{\omega \rightarrow 0} \left(\frac{1}{1 + \omega^2} \right) = -1$$


$$\lim_{\omega \rightarrow 0} \operatorname{Im}[G(j\omega)] = -\lim_{\omega \rightarrow 0} \left(\frac{1}{\omega + \omega^3} \right) = -\infty$$

Stabilità Sistemi Lineari Reazionati

Utilizziamo la funzione “**horner**” al fine di tracciare il **diagramma polare** di **$G(j\omega)$** e della simmetrica **$G(-j\omega)$** , relativo all'intervallo della **pulsazione ω** dato da **$(0,1 \text{ r/s} < \omega < 1000 \text{ r/s})$** ed approntiamo, in ambiente **SCILAB**, il seguente insieme di istruzioni con lo scopo di verificare l'esistenza dello **asintoto** relativo alla **parte reale**:

```
--> num=1;  
--> den=poly([0 1], 's', 'c')*poly([1 1], 's', 'c');  
--> w=logspace(-1,3,500);  
--> RFP= horner(num, %i*w)./(horner(den, %i*w));  
--> RFN= horner(num, -%i*w)./(horner(den, -%i*w));  
--> plot(real(RFP), imag(RFP), real(RFN), imag(RFN))
```

Stabilità Sistemi Lineari Reazionati



Stabilità Sistemi Lineari Reazionati


Verifichiamo l'andamento del **diagramma polare $G(j\omega)$** e del suo **simmetrico $G(-j\omega)$** , ottenuto in precedenza per la **Funzione di Trasferimento $G(s)$** , utilizzando l'istruzione relativa al tracciamento dei **diagrammi di NYQUIST**.

In ambiente SCILAB, il listato che segue:

```
--> num=1;  
--> den=poly([0 1],'s','c')*poly([1 1],'s','c');  
--> giesse=syslin('c',num,den);  
--> nyquist(giesse)
```

genera il grafico relativo al diagramma di Nyquist


Stabilità Sistemi Lineari Reazionati



Stabilità Sistemi Lineari Reazionati

Si analizzi la **stabilità** del **sistema** di **controllo** avente **Funzione** di **Trasferimento** d'anello **$L(s)$** al variare di **T** .


$$L(s) = \frac{1}{s \cdot (1 + sT)^2}$$



```
--> num=1;  
--> T=4; den1=poly([0 1],'s','c')*poly([1 T],'s','c')^2;  
--> elle1=syslin('c',num,den1);  
--> T=2; den2=poly([0 1],'s','c')*poly([1 T],'s','c')^2;  
--> elle2=syslin('c',num,den2);  
--> T=0.5; den3=poly([0 1],'s','c')*poly([1 T],'s','c')^2;  
--> elle3= syslin('c',num,den3);  
--> bode([elle1;elle2;elle3])
```

Stabilità Sistemi Lineari Reazionati

Si raffrontano i tre diagrammi di BODE dei moduli e delle fasi



Stabilità Sistemi Lineari Reazionati

Dall'analisi dei tre **diagrammi** di **BODE** e con riferimento allo omonimo **criterio** di **stabilità** di **BODE**, si deduce quanto segue:

$$L_1(s) = \frac{1}{s \cdot (1 + 4s)^2}$$

$$\omega_C = 0,32 \text{ rad/sec} \Rightarrow f_C = 0,051 \text{ Hz}$$

$$\Phi_C = -194^\circ \Rightarrow \Phi_M < 0$$

Sistema retroazionato instabile

$$L_2(s) = \frac{1}{s \cdot (1 + 2s)^2}$$

$$\omega_C = 0,50 \text{ rad/sec} \Rightarrow f_C = 0,0637 \text{ Hz}$$

$$\Phi_C = -180^\circ \Rightarrow \Phi_M = 0$$

Sistema al limite della stabilità

$$L_3(s) = \frac{1}{s \cdot (1 + 0,5s)^2}$$

$$\omega_C = 0,848 \text{ rad/sec} \Rightarrow f_C = 0,135 \text{ Hz}$$

$$\Phi_C = -136^\circ \Rightarrow \Phi_M = 44^\circ > 0$$

Sistema reazionato asintoticamente stabile

Stabilità Sistemi Lineari Reazionati

Verifichiamo per via **analitica** quanto già ottenuto con **Scilab**

Il **sistema retroazionato** si troverà al **limite della stabilità** quando la sua **fase critica** Φ_C , calcolata alla **pulsazione critica** ω_C , varrà $\Phi_C = -180^\circ$, ovvero la **fase margine** Φ_M sarà $\Phi_M = 0^\circ$.

$$\angle L(j\omega_C) = -180^\circ \Rightarrow \angle L(j\omega_C) = -\angle(j\omega) - 2 \cdot [\angle(1+j\omega_C T)] = -180^\circ$$

$$-90^\circ - 2 \cdot \arctan(\omega_C T) = -180^\circ \Rightarrow 2 \cdot \arctan(\omega_C T) = 90^\circ$$

$$\text{consegue che: } \arctan(\omega_C T) = 45^\circ \Rightarrow \omega_C T = \tan(45^\circ) \Rightarrow \omega_C T = 1$$

Alla pulsazione critica ω_C il modulo della funzione d'anello $L(j\omega_C)$ deve soddisfare la condizione $|L(j\omega_C)| = 1$, il che implica quanto segue:

$$|L(j\omega_C)| = 1 / |j\omega_C \cdot (1+j\omega_C T)^2| = 1 \Rightarrow 1 / \omega_C \cdot [1 + (\omega_C T)^2] = 1$$


$$\frac{1}{(1/T) \cdot (1+1)} = 1 \Rightarrow \frac{T}{2} = 1 ; \text{ ovvero: } T = 2 \text{ sec, da cui: } \omega_C = 0,5 \text{ r/s}$$

Stabilità Sistemi Lineari Reazionati

Confermiamo mediante il **criterio** di **stabilità** di **NYQUIST** le conclusioni già ottenute col ricorso al **criterio** di **stabilità** di **BODE**

Si analizza, pertanto, in **ambiente SCILAB**, la **stabilità** del **sistema** di **controllo** avente **Funzione** di **Trasferimento** d'anello **L(s)**, al variare di **T**, adottando il **criterio** di **stabilità** di **Nyquist**.

$$L(s) = \frac{1}{s \cdot (1 + sT)^2}$$




L'istruzione che soddisfa allo scopo, in **ambiente Scilab**, assume la forma seguente:

--> nyquist([elle1;elle2;elle3])


a cui corrisponde il tracciamento dei tre **diagrammi** di **Nyquist**

Stabilità Sistemi Lineari Reazionati



Stabilità Sistemi Lineari Reazionati

Dai precedenti **diagrammi** di Nyquist non viene bene evidenziato che la **fase**, per $\omega \rightarrow \infty$, assume il valore di $\Phi = -270^\circ$. Usiamo l'istruzione: **--> nyquist([elle1;elle2;elle3],0.05,500)** al fine di ottenere:



Il **punto $(-1, j0)$** è **esterno** al grafico di Nyquist di **elle3**
Sistema reazionato **STABILE** asintot.

Il **punto $(-1, j0)$** si trova **sul** grafico di Nyquist di **elle2**
Sistema reazionato **al limite della STABILITÀ**


Il **punto $(-1, j0)$** è **interno** al grafico di Nyquist di **elle1**
Sistema reazionato **INSTABILE**

Stabilità Sistemi Lineari Reazionati

Si vuole verificare tramite il **criterio di Nyquist** l'instabilità del sistema reazionato con funzione di trasferimento d'anello $L(s)$ e costante di tempo $T=4$.


Cammino di Nyquist:

La singolarità nell'origine viene evitata da una semicirconferenza di raggio infinitesimo percorsa in senso antiorario. Così, si ottiene: $P=0$ (zero poli contenuti nel cammino)




$$L(s) = \frac{1}{s \cdot (1 + 4s)^2}$$

```
-->num=poly([1 0],'s','c');  
-->den=poly([0 1],'s','c')*poly([1 4],'s','c')^2;  
-->ellesse=syslin('c',num,den);  
-->w=[(-0.03:-0.001:-8.2) (8.2:-0.001:0.03)];  
-->nyquist(ellesse,w)
```



Stabilità Sistemi Lineari Reazionati



$$N = -2$$

$$P_P = 0$$


$$N \neq P_P$$

Sistema
reazionato
INSTABILE

Stabilità Sistemi Lineari Reazionati

Dato il sistema di controllo con funzione di trasferimento d'anello $L(s)$, si trovi il **valore** di K che dà un **margin**e di fase $\Phi_M = 45^\circ$ e si determini la relativa **pulsazione critica** ω_C .

$$L(s) = \frac{K}{s \cdot (1 + s)^2}$$




Si inizia col porre $K = 1$ e si ricava il corrispondente **diagramma** di **Bode** al fine di valutare la **pulsazione critica** ω_C e la **fase critica** Φ_C

```
--> num=1;  
--> den=poly([0 1],'s','c')*poly([1 1],'s','c')^2;  
--> ellesse=syslin('c',num,den);  
--> bode(ellesse)
```

Stabilità Sistemi Lineari Reazionati

Si ottiene il seguente diagramma di Bode relativo a $K=1$



Stabilità Sistemi Lineari Reazionati

Si nota che il **diagramma** di **Bode** del **modulo** di **L(s)** **taglia** l'**asse** a **zero db**, approssimativamente, alla **frequenza critica** **f_C** di poco superiore a **0,11 Hz**; a tale frequenza, la **fase critica** vale approssimativamente **Φ_C = -160°** a cui corrisponde una:

$$\text{fase margine: } \Phi_M = 180^\circ - |\Phi_C| = 180^\circ - 160^\circ = 20^\circ$$

Al fine di ottenere **informazioni di maggior rigore** conviene fare ricorso all'uso delle **istruzioni: p_margin() e g_margin()**

--> **[phicr frecr] = p_margin(ellesse)**

freocr = 0.1085958 \Rightarrow **(frequenza critica f_C)**

phicr = -158.61361 \Rightarrow **(fase critica Φ_C)**

--> **[gainmar frepigr] = g_margin(ellesse)**

frepigr = 0.1085958 \Rightarrow **(frequenza f_π relativa a Φ = π)**

gainmar = 6.0205999 \Rightarrow **(guadagno margine G_M in dB)**

Stabilità Sistemi Lineari Reazionati

Dall'analisi del **diagramma di Bode** si evince la **necessità** di **abbassare il diagramma dei moduli** onde **limitare**, sia la **pulsazione critica**, sia la **fase critica**, in modo d'**aumentare** la **fase margine**.


In particolare, affinché la **fase critica** si porti a $\Phi_C = -135^\circ$, il **diagramma dei moduli** deve **tagliare** l'**asse a zero dB** ad una **pulsazione critica inferiore** e deve **abbassarsi** di circa **6,5 dB**, fatte salve le **approssimazioni** afferenti la **precisione** con cui si effettuano le **letture** sul **grafico**.

Si conclude, pertanto, come segue:

$$20 \cdot \log_{10} k = -6,5 \quad \Rightarrow \quad \log_{10} k = -0,325$$

$$k = 10^{-0,325} \quad \Rightarrow \quad k = 0,473$$


Stabilità Sistemi Lineari Reazionati



Stabilità Sistemi Lineari Reazionati

In ambiente SCILAB, ridefiniamo la **funzione di trasferimento d'anello** $L(s)$ con $k=0,473$ al fine di ottenere il **diagramma di BODE modificato**, in coerenza con tutte le considerazioni appena effettuate e verificare la loro adeguatezza.


$$L(s) = \frac{0,473}{s \cdot (1 + s)^2}$$



```
--> K=0,473;  
--> den=poly([0 1],'s','c')*poly([1 1],'s','c')^2;  
--> ellesse2=syslin('c',K,den);  
--> [phicr2 frecr2] = p_margin(ellesse2);  
--> [gainmar2 frepigr2] = g_margin(ellesse2);  
--> bode(ellesse2)
```

Stabilità Sistemi Lineari Reazionati

Diagramma di BODE relativo al valore di $k=0,473$



$$f_C = 0,06465 \text{ Hz}$$

$$\omega_C = 2\pi f_C$$

$$\omega_C = 0,407 \text{ r/s}$$

$$\Phi_C = -134,20^\circ$$

$$f_\pi = 0,159 \text{ Hz}$$

$$\omega_\pi = 2\pi f_\pi$$

$$\omega_\pi = 1,00 \text{ r/s}$$

$$G_M = 12,52 \text{ dB}$$

Stabilità Sistemi Lineari Reazionati

Procedura analitica

Una **fase margine** $\Phi_M = 45^\circ$ richiede una **fase critica** pari a $\Phi_C = -135^\circ$; pertanto, alla **pulsazione critica** ω_C deve **essere soddisfatta** la seguente **relazione angolare**:

$$\begin{aligned}\angle L(j\omega_c) &= \angle k - (\pi/2) - 2 \cdot \arctan \omega_c && \text{con } k > 0 \text{ (positivo)} \\ -135^\circ &= -(\pi/2) - 2 \cdot \arctan \omega_c && \Rightarrow -45^\circ = -2 \cdot \arctan \omega_c\end{aligned}$$


ovvero: $\arctan \omega_c = 22,5^\circ \Rightarrow \omega_c = \tan 22,5^\circ = 0,414 \text{ r/s}$

Alla **pulsazione critica** $\omega_c = 0,414 \text{ rad/sec}$, il **modulo** della **funzione di trasferimento d'anello** vale $|L(j\omega_c)| = 1$; cioè, **deve essere soddisfatta** la seguente **condizione**:

$$|L(j\omega_c)| = 1 \Rightarrow 1 = \left| \frac{K}{j\omega_c \cdot (1 + j\omega_c)^2} \right| \Rightarrow \omega_c \cdot (1 + \omega_c^2) = K$$

ovvero: $k = 0,485 \Rightarrow k_{dB} = 20 \log_{10} 0,485 = -6,3 \text{ dB}$

Analisi di Sensitività e Sensitività Complementare



$$\begin{pmatrix} Y(s) \\ U(s) \\ E(s) \end{pmatrix} = \begin{pmatrix} F(s) & S(s) & -F(s) \\ Q(s) & -Q(s) & -Q(s) \\ S(s) & -S(s) & N(s) \end{pmatrix} \bullet \begin{pmatrix} Y_{sp}(s) \\ D(s) \\ N(s) \end{pmatrix}$$

Analisi di Sensitività e Sensitività Complementare

Le **FUNZIONI** di **SENSITIVITÀ** definiscono le funzioni di trasferimento tra gli **ingressi ritenuti significativi** ai fini del controllo e le **uscite di interesse** del sistema di controllo.

Le **uscite di interesse** ai fini del controllo sono:

$e(t)$ = errore di inseguimento;

$y(t)$ = uscita controllata;

$u(t)$ = azione di controllo

Gli **ingressi ritenuti significativi** ai fini del controllo sono:

$n(t)$ = disturbo di misura;

$d(t)$ = disturbo in uscita e dell'attuatore;

$y_{Sp}(t)$ = segnale di riferimento – Set Point

Analisi di Sensitività e Sensitività Complementare

Le **FUNZIONI di SENSITIVITÀ** sono, pertanto, definite dalle **funzioni di trasferimento** che di seguito si esplicitano:

➤ **FUNZIONE di SENSITIVITÀ S(s)**

$$S(s) = \frac{1}{1 + R(s)G(s)} = \frac{Y(s)}{D(s)} = \frac{E(s)}{Y_{sp}(s)} = -\frac{E(s)}{D(s)}$$

➤ **FUNZIONE di SENSITIVITÀ COMPLEMENTARE F(S)**


$$F(s) = \frac{R(s)G(s)}{1 + R(s)G(s)} = \frac{Y(s)}{Y_{sp}(s)} = -\frac{Y(s)}{N(s)} = \frac{E(s)}{N(s)}$$

➤ **FUNZIONE di SENSITIVITÀ del CONTROLLO Q(S)**

$$Q(s) = \frac{R(s)}{1 + R(s)G(s)} = \frac{U(s)}{Y_{sp}(s)} = -\frac{U(s)}{D(s)} = -\frac{U(s)}{N(s)}$$

Analisi di Sensitività e Sensitività Complementare

Sia assegnato il sistema avente **Funzione di Trasferimento ad anello aperto $L(s)$** , come di seguito riportato:



Si traccino i diagrammi di **BODE** associati alle **Funzioni di Trasferimento $S(s)$ ed $F(s)$** osservandone l'andamento.

$$S(s) = \frac{1}{1 + L(s)}$$

$$F(s) = \frac{L(s)}{1 + L(s)}$$

Analisi di Sensitività e Sensitività Complementare

Si tratta di esaminare un sistema di controllo in retroazione nel quale si considera $\mathbf{R(s)} = \mathbf{1}$ e rumore di misura $\mathbf{n(t)} = \mathbf{0}$.

In ambiente SCILAB si stendono le seguenti righe di codice:

Funzione di trasferimento S(s)


```
s=poly(0,'s');  
num=1;den=s*(1+s);  
ellesse=syslin('c',num,den);  
esse=1/(1+ellesse);  
w=logspace(-2,1,2000);  
scf(1),bode(esse,w)
```

Funzione di trasferimento F(s)

```
s=poly(0,'s');  
num=1;den=s*(1+s);  
ellesse=syslin('c',num,den);  
effe=ellesse/(1+ellesse);  
w=logspace(-2,1,2000);  
scf(2),bode(effe,w)
```


al fine di ottenere i **diagrammi** di **BODE** del Modulo e della Fase delle Funzioni di Trasferimento **S(s)** e **F(s)**.

Analisi di Sensività e Sensività Complementare



$S(s)$

Analisi di Sensitività e Sensitività Complementare



$F(s)$

Analisi di Sensitività e Sensitività Complementare

Si traccino, confrontandoli, i grafici delle risposte allo scalino delle Funzioni di Trasferimento $S(s)$ e $F(s)$, sia nell'ambiente SCILAB, sia in ambiente SCICOS.

In ambiente SCILAB si stendono le seguenti righe di codice:

Funzione di trasferimento $S(s)$


```
s=poly(0,'s');  
num=1;den=s*(1+s);  
ellesse=syslin('c',num,den);  
esse=1/(1+ellesse);  
t=0:0.01:12;  
yesse=csim('step',t,esse);  
plot(t,[yesse;yeffe),xgrid
```

Funzione di trasferimento $F(s)$

```
s=poly(0,'s');  
num=1;den=s*(1+s);  
ellesse=syslin('c',num,den);  
effe=ellesse/(1+ellesse);  
t=0:0.01:12;  
yeffe=csim('step',t,effe);
```

Si rilevino il valore limite per $t \rightarrow \infty$ delle due risposte nonché i rispettivi tempi di assestamento, di sovra elongazione e sotto elongazione.

Analisi di Sensività e Sensività Complementare




Analisi di Sensitività e Sensitività Complementare

Simuliamo in ambiente **SCICOS** le **Funzioni di Trasferimento** **S(s)** ed **F(s)** al fine pervenire agli **stessi risultati** già conseguiti in ambiente **SCILAB**


$$S(s) = \frac{1}{1 + L(s)} = \frac{1}{1 + \frac{1}{s \cdot (s + 1)}} = \frac{s \cdot (s + 1)}{1 + s + s^2} = S(s)$$

$$F(s) = \frac{L(s)}{1 + L(s)} = \frac{\frac{1}{s \cdot (1 + s)}}{1 + \frac{1}{s \cdot (1 + s)}} = \frac{1}{1 + s + s^2} = F(s)$$

Analisi di Sensività e Sensitività Complementare




Analisi di Sensività e Sensività Complementare



Analisi di Sensitività e Sensitività Complementare

Si consideri ora lo schema a blocchi, riportato nella figura, in cui si considera un **disturbo di misura $N(s)$** presente sulla linea di reazione. Sia $L(s)$ la funzione di trasferimento d'anello


$$L(s) = \frac{1}{s \cdot (1 + s)}$$



Si determini la Funzione di Trasferimento fra il **rumore $N(s)$** e l'**uscita $Y(s)$** e si decida di simulare sia in ambiente **SCILAB**, sia in ambiente **SCICOS**, la risposta del **sistema retroazionato** ad una **variazione a scalino del disturbo $n(t)$** .

Analisi di Sensitività e Sensitività Complementare

In ambiente **SCILAB**, considerato $Y_{SP}=0$, la Funzione di Trasferimento da $N(s)$ ad $Y(s)$ è così calcolabile:



$$Y(s) = L(s) \cdot \{Y_{sp}(s) - [N(s) + Y(s)]\}$$

$$Y(s) = L(s) \cdot \{0 - [N(s) + Y(s)]\}$$

$$Y(s) = -L(s)N(s) - L(s)Y(s)$$

$$[1 + L(s)] \cdot Y(s) = -L(s)N(s)$$

$$\frac{Y(s)}{N(s)} = \frac{-L(s)}{1 + L(s)} = -F(s)$$

Analisi di Sensitività e Sensitività Complementare


Per ottenere la risposta ad un **disturbo $n(t)$** a **gradino unitario**, in ambiente **SCILAB** si procede con lo stendere le righe di codice che a lato si riportano:

Funzione di trasferimento -F(s)

```
s=poly(0,'s');  
num=1;den=s*(1+s);  
ellesse=syslin('c',num,den);  
giesse=-ellesse/(1+ellesse);  
t=0:0.01:12;  
yenne=csim('step',t,giesse);  
plot(t,yenne),xgrid
```

$$\frac{Y(s)}{N(s)} = \frac{-L(s)}{1+L(s)} = \frac{-\frac{1}{s \cdot (1+s)}}{1 + \frac{1}{s \cdot (1+s)}} = \frac{-1}{1+s+s^2} = -F(s)$$

Analisi di Sensitività e Sensitività Complementare



Analisi di Sensitività e Sensitività Complementare

In ambito SCICOS la simulazione è relativa allo schema seguente:

